

**stichting
mathematisch
centrum**



DEPARTMENT OF NUMERICAL MATHEMATICS

NOVEMBER

NUMAL, A LIBRARY OF NUMERICAL PROCEDURES IN ALGOL 60

VOLUME 7. INTERPOLATION AND APPROXIMATION

NW

2e boerhaavestraat 49 amsterdam

SECTION : 7.1

(SEPTEMBER 1974)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: MATHEMATICAL CENTRE.

RECEIVED: 730618.

BRIEF DESCRIPTION:

NEWTON CALCULATES THE COEFFICIENTS OF THE NEWTON POLYNOMIAL THROUGH GIVEN INTERPOLATION POINTS AND CORRESPONDING FUNCTION VALUES.

KEYWORDS:

NEWTON INTERPOLATION,
POLYNOMIAL COEFFICIENTS,
DIVIDED DIFFERENCES.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:
"PROCEDURE" NEWTON(N,X,F);
"VALUE"N;"INTEGER"N;"ARRAY"X,F;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;
THE DEGREE OF THE POLYNOMIAL;
X: <ARRAY IDENTIFIER>;
"ARRAY"X[0:N];
THE INTERPOLATION POINTS;

F: <ARRAY IDENTIFIER>;
"ARRAY"F[0:N];

ENTRY: THE POLYNOMIAL VALUES AT THE INTERPOLATION POINTS;
EXIT: THE COEFFICIENTS OF THE NEWTON POLYNOMIAL.

PROCEDURES USED: NONE.

RUNNING TIME: THE NUMBER OF DIVISIONS IS $N(N+1)/2$.

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE:

THE COEFFICIENTS OF THE NEWTON POLYNOMIAL ARE CALCULATED BY INTERPOLATION AT THE GIVEN ARGUMENTS AND FUNCTION VALUES; THE RESULTING SET OF EQUATIONS IS SOLVED BY TRANSFORMING THE CORRESPONDING LOWER TRIANGULAR MATRIX TO DIAGONAL FORM.

EXAMPLE OF USE:

```
"BEGIN" "ARRAY" X,F[0:2];
  "PROCEDURE" NEWTON(N,X,F);
  "CODE" 36010;
    X[0]:=0;X[1]:=5;X[2]:=1;
    F[0]:=1;F[1]:=F[2]:=0;
    NEWTON(2,X,F);
    OUTPUT(61,"(", "(", "THE NEWTON COEFF. ARE")",
      /,3(N)"/",F[0],F[1],F[2]);
  "END" TSTNEWTON;
```

```
THE NEWTON COEFF. ARE
+1.0000000000000000"+000 -2.0000000000000000"+000 +2.0000000000000000"+000
```

SOURCE TEXT(S):

```
"CODE" 36010;
"PROCEDURE" NEWTON(N,X,F);
"VALUE" N; "INTEGER" N; "ARRAY" X,F;
"COMMENT" NEWTON DETERMINES THE COEFFICIENTS C[J], J=0,...,N,
OF THE INTERPOLATION POLYNOMIAL C[0] + C[1] *(X-X[0]) + ... +
C[N] * (X-X[0]) * ... * (X-X[N-1]) OUT OF N+1 LIN. EQUAT.
THE ARGUMENTS AND FUNCTION VALUES MUST BE GIVEN IN
ARRAY X, F[0:N]. THE ARRAY F IS OVERWRITTEN BY
THE COEFFICIENTS C[J], J=0,...,N;
"BEGIN" "INTEGER" K,I,IM1;
"REAL" XIM1,FIM1;
IM1:=0;
"FOR" I:=1 "STEP" 1 "UNTIL" N "DO"
  "BEGIN" FIM1:=F[IM1];XIM1:=X[IM1];
    "FOR" K:=I "STEP" 1 "UNTIL" N "DO" F[K]:= (F[K]-FIM1)/(X[K]-XIM1);
    IM1:=I
  "END"
"END"
"END" NEWTON;
"EOP"
```


AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: MATHEMATICAL CENTRE.

RECEIVED: 730831.

BRIEF DESCRIPTION:

THIS SECTION CONTAINS THREE PROCEDURES:

INI SELECTS A (SUB)SET OF INTEGERS OUT OF A GIVEN SET OF INTEGERS.

SNDREMEZ EXCHANGES AT MOST $N+1$ NUMBERS WITH NUMBERS OUT OF A REFERENCE SET.

MINMAXPOL CALCULATES THE COEFFICIENTS OF THE POLYNOMIAL (IN THE GRUNERT FORM) THAT APPROXIMATES A FUNCTION, GIVEN FOR DISCRETE ARGUMENTS, IN SUCH A WAY THAT THE INFINITY NORM OF THE ERROR VECTOR IS MINIMISED.

KEYWORDS:

ARGUMENTS EXTREME VALUES,
CHEBYSHEV POLYNOMIAL,
SECOND REMEZ ALGORITHM,
MINIMAX POLYNOMIAL APPROXIMATION.

SUBSECTION : INI.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

"PROCEDURE" INI(N,M,S);
"VALUE" N,M;"INTEGER" N,M""INTEGER""ARRAY"S;
"CODE" 36020;

THE MEANING OF THE FORMAL PARAMETERS IS:
N,M:<ARITHMETIC EXPRESSION>;

THE NUMBER OF POINTS TO BE SELECTED EQUALS $N+1$;
THE REFERENCE SET CONTAINS THE NUMBERS $0,1,\dots,M,M \geq N$;
S: <ARRAY IDENTIFIER>;
"INTEGER" "ARRAY" S[0:N];
EXIT: THE SELECTED INTEGERS ARE DELIVERED IN S.

PROCEDURES USED: NONE.

SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 2

LANGUAGE: ALGOL 60.

METHODS AND PERFORMANCE:

THE ARGUMENTS FOR WHICH THE CHEBYSHEV POLYNOMIAL OF DEGREE N ATTAINS ITS EXTREME VALUES ON THE INTERVAL $[-1,1]$ ARE TRANSFORMED TO THE INTERVAL $[0,M]$ BY A LINEAR TRANSFORMATION; FINALLY THE NUMBERS ARE PROPERLY ROUNDED.

REFERENCES : SEE MINIMAXPOL (THIS SECTION).

EXAMPLE OF USE:

```
"BEGIN""INTEGER""ARRAY"S[0:2];
"PROCEDURE"INI(N,M,S);"CODE"36020;
INI(2,20,S);
OUTPUT(61,"("{"INI SELECTS OUT OF 0,1,...,20 THE NUMBERS:"}"/,
3(B-ZDB)"")",S[0],S[1],S[2])
"END"
```

```
INI SELECTS OUT OF 0,1,...,20 THE NUMBERS:
0    10    20
```

SUBSECTION : SNDREMEZ.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

```
"PROCEDURE"SNDREMEZ(N,M,S,G,EM);
"VALUE"N,M;"INTEGER"N,M;"INTEGER""ARRAY"S;"ARRAY" G,EM;
"CODE" 36021;
```

THE MEANING OF THE FORMAL PARAMETERS IS:

N,M: <ARITHMETIC EXPRESSION>;

THE NUMBER OF POINTS TO BE EXCHANGED IS SMALLER THAN OR EQUAL TO $N+1$; THE REFERENCE SET CONTAINS THE NUMBERS $0,1,\dots,M$, $M \geq N$;

S: <ARRAY IDENTIFIER>;

"INTEGER" "ARRAY" S[0:N];

ENTRY: IN S ONE MUST GIVE $N+1$ (STRICTLY)MONOTONE INCREASING NUMBERS OUT OF $0,\dots,M$;EXIT : $N+1$ (STRICTLY) MONOTONE INCREASING NUMBERS OUT OF THE NUMBERS $0,1,\dots,M$;

G: <ARRAY IDENTIFIER>;

"ARRAY" G[0:M];

ENTRY: IN ARRAY G[0:M] ONE MUST GIVE FUNCTIONVALUES;

EM: <ARRAY IDENTIFIER>;

"ARRAY" EM[0:1];

ENTRY: $0 < EM[0] \leq G[I], I=0,\dots,M$;EXIT: $EM[1] := \text{INFINITY NORM OF ARRAY } G[0:M]$.

SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 3

PROCEDURES USED: ABSMAXVEC=CP31060.

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE:

THE SECOND REMEZ ALGORITHM IS USED (MEINARDUS, G. (1964)).

REFERENCES : SEE MINMAXPOL (THIS SECTION).

EXAMPLE OF USE:

```
"BEGIN""ARRAY"EM[0:1],G[0:7];"INTEGER""ARRAY"S[0:2];
"PROCEDURE"SNOREMEZ(N,M,S,G,EM);"CODE"36021;
G[0]:=10;G[1]:=12;G[2]:=-15;G[3]:=-10;
G[4]:=-14;G[5]:=15;G[6]:=10;G[7]:=11;
EM[0]:=10;S[0]:=0;S[1]:=3;S[2]:=6;
OUTPUT(61,"("("THE NUMBERS:""),/,("S[J]:""),3(B-D),/,
      ("G[S[J]]:""),3(B-DD)"),",
      S[0],S[1],S[2],G[S[0]],G[S[1]],G[S[2]]);
SNOREMEZ(2,7,S,G,EM);
OUTPUT(61,"("//,("ARE EXCHANGED WITH:""),/,("S[J]:""),3(B-D),/,
      ("G[S[J]]:""),3(B-DD),//,
      ("THE REFERENCE SET OF FUNCTIONVALUES IS:""),/,8(B-DD)"),",
      S[0],S[1],S[2],G[S[0]],G[S[1]],G[S[2]],
      G[0],G[1],G[2],G[3],G[4],G[5],G[6],G[7])
"END"
```

THE NUMBERS:

```
S[J]:  0  3  6
G[S[J]]: 10 -10 10
```

ARE EXCHANGED WITH:

```
S[J]:  0  2  5
G[S[J]]: 10 -15 15
```

THE REFERENCE SET OF FUNCTIONVALUES IS:

```
10 12 -15 -10 -14 15 10 11
```


SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 4

SUBSECTION : MINMAXPOL.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

"PROCEDURE"MINMAXPOL(N,M,Y,FY,CO,EM);

"VALUE"N,M;"INTEGER"N,M;"ARRAY"Y,FY,CO,EM;

"CODE" 36022;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;

THE DEGREE OF THE APPROXIMATING POLYNOMIAL IS N-1;

M: <ARITHMETIC EXPRESSION>;

THE NUMBER OF REFERENCE FUNCTION VALUES VIZ. ARGUMENTS IS M+1;

Y,FY: <ARRAY IDENTIFIERS>;

"ARRAY"Y,FY[0:M];

ENTRY: FY[I] IS THE FUNCTION VALUE AT Y[I], FOR I=0,...M;

CO: <ARRAY IDENTIFIER>;

"ARRAY"CO[0:N];

EXIT: THE COEFFICIENTS OF THE APPROXIMATING POLYNOMIAL, OF DEGREE N-1, ARE DELIVERED IN CO[0:N-1] (CO[N-1] IS COEFFICIENT OF $Y^{(N-1)}$);

EM: <ARRAY IDENTIFIER>;

"ARRAY"EM[0:3];

ENTRY: EM[2]: THE MAXIMUM ALLOWED NUMBER OF ITERATIONS;

EXIT: EM[0]: THE DIFFERENCE OF THE GIVEN FUNCTION AND THE POLYNOMIAL IN THE FIRST APPROXIMATION POINT;

EM[1]: THE INFINITY NORM OF THE ERROR OF APPROXIMATION OVER THE DISCRETE INTERVAL;

EM[3]: THE NUMBER OF ITERATIONS PERFORMED.

PROCEDURES USED:

ELMVEC =CP34020,

ABSMAXVEC=CP31060,

NEWTON =CP36010,

POL =CP31040,

NEWGRN =CP31050,

INI =CP36020,

SNDREMEZ =CP36021.

REQUIRED CENTRAL MEMORY : ONE INTEGER ARRAY AND TWO REAL ARRAYS OF ORDER $N + 1$, AND ONE REAL ARRAY OF ORDER $M + 1$ ARE USED.

RUNNING TIME:

THE SECOND REMEZ ALGORITHM(ON A DISCRETE SET) IS QUADRATIC CONVERGENT; IN EACH ITERATION THE NUMBER OF OPERATIONS (MULTIPLICATIONS AND ADDITIONS) IS PROPORTIONAL TO $M \cdot N$.

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE: SEE REF. [1], C4.7.

REFERENCES:

- [1]. G. MEINARDUS.
APPROXIMATION OF FUNCTIONS AND THEIR NUMERICAL TREATMENT
(GERMAN). SPRINGER TRACTS IN NATURAL PHILOSOPHY.
VOLUME 4. 1964.

EXAMPLE OF USE:

```

"BEGIN""INTEGER"N;
"PROCEDURE"MINMAXPOL(N,M,Y,FY,CO,EM); "CODE" 36022;

  "PROCEDURE" COMPUTE(N,A,B,F);
  "VALUE" N,A,B;"INTEGER" N;"REAL" A,B;
  "REAL" "PROCEDURE" F;
  "BEGIN" "INTEGER" K,L,M;
  "REAL"R,T,IDM;
  "ARRAY" COEF[0:N],EM[0:3];
  EM[2]:=10*N;
  M:=100*N;
  "BEGIN" "ARRAY" Y,FY[0:M];
  IDM:=(B-A)/M;
  R:=Y[0]:=A;FY[0]:=F(R);
  R:=Y[M]:=B;FY[M]:=F(R);
  L:=M-1;
  "FOR"K:=1"STEP"1"UNTIL"L"DO"
  "BEGIN"R:=Y[K]:=A+K*IDM;FY[K]:=F(R) "END";
  MINMAXPOL(N,M,Y,FY,COEF,EM);
  OUTPUT(61,"("COEF:""/")");
  "FOR"K:=0"STEP"1"UNTIL"N-1"DO"OUTPUT(61,"(" ",COEF[K]);
  OUTPUT(61,"("/8S/,2(N),2(B+3ZDB),/""",("EM[0:3]""",EM[0],EM[1],
  EM[2],EM[3]));
  "END";
  "END" COMPUTE;

"REAL""PROCEDURE"F(X);"VALUE"X;"REAL"X;
F:=1/(X-10);

  "FOR" N:=2"DO"
  "BEGIN" OUTPUT(61,"("//,("DEGREE=")",D//")",N-1);
    COMPUTE(N,-1,1,F)
  "END"
  "END";

```

DEGRE E=1

```

COEF:
-1.0050378153393"-001  -1.0101010101010"-002
EM[0:3]
-5.0631947616870"-004  +5.0631947616870"-004      +20      +3

```


SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 6

SOURCE TEXT(S) :

```

"CODE"36020;
"PROCEDURE" INI(N,M,S);
"VALUE" N,M;"INTEGER" N,M;
"INTEGER" "ARRAY" S;
"COMMENT" INI DELIVERS (MONOTONE) THE ROUNDED VALUES
OF THE ARGUMENTS,WHERE THE CHEBYSHEV POLYNOMIAL
OF DEGREE N(TRANSFORMED TO THE INTERVAL [0,M],M>=N)
ATTAINS ITS MAXIMUM VALUES,
IN INTEGER ARRAY S[0:N];
"BEGIN""INTEGER" I,J,K,L;"REAL" PIN2;
    PIN2:=ARCTAN(1)*2/N;
    K:=0;L:=N-1;J:=S[0]:=0;S[N]:=M;
    "FOR" K:=K+1 "WHILE" K < L "DO"
        "BEGIN" I:=SIN(K*PIN2)**2*M;
            J:=S[K]:="IF" I<=J "THEN" J+1 "ELSE" I;
            S[L]:=M-J;L:=L-1
        "END" K;
    "IF" L*2=N "THEN" S[L]:=M/2;
    "END" INI;
    "EOP"

"CODE"36021;
"PROCEDURE" SNDREMEZ(N,M,S,G,EM);
"VALUE" N,M;"INTEGER" N,M;
"INTEGER" "ARRAY" S;"ARRAY" G,EM;
"COMMENT" SNDREMEZ EXCHANGES ATMOST N+1 NUMBERS ,GIVEN IN
INTEGER ARRAY S[0:N], WITH NUMBERS OUT OF THE
REFERENCE SET 0,...M, UNDER THE CONDITIONS:
I. THE ALTERNANCE PROPERTY OF THE FUNCTIONVALUES G(S[J]),
J=0,...N IS PRESERVED.
II. !G(S[J])!>=!EM[0]!,J=0,...N.
III. THE FIRST INDEX K , WITH G[K]=INFINITY NORM OF G,
IS ONE OF THE RESULTING NUMBERS S[0],...S[N].
IN ARRAY G[0:M] ONE MUST GIVE ERROR FUNCTION VALUES.
MOREOVER,
    EM[1]:=INFINITY NORM OF G,
    THE PROCEDURE ABSMAXVEC IS USED;
"BEGIN" "INTEGER" S0,SN,SJP1,I,J,K,UP,INDEXMAX,LOW,NM1;
    "REAL" MAX,MSJP1,HI,HJ,HE,ABSE,H;
    "REAL" "PROCEDURE" ABSMAXVEC(K,LOW,UP,A); "CODE" 31060;

```

"COMMENT"

SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 7
;

```

INDEX MAX:=S0:=SJP1:=S[0];
HE:=EM[0];LOW:=S0+1;
MAX:=MSJP1:=ABSE:=ABS(HE);
NM1:=N-1;
"FOR" J:= 0 "STEP" 1 "UNTIL" NM1 "DO"
"BEGIN"
  UP:= S[J+1]-1;
  H:= ABSMAXVEC(I,LOW,UP,G);
  "IF" H > MAX "THEN" "BEGIN" MAX:= H; INDEX MAX:= I "END";
  "IF" H > ABSE "THEN"
    "BEGIN" "IF" HE * G[I] > 0 "THEN"
      "BEGIN" S[J]:= "IF" MSJP1 < H "THEN" I "ELSE" SJP1;
      SJP1:= S[J+1]; MSJP1:= ABSE
    "END" "ELSE"
      "BEGIN" S[J]:= SJP1; SJP1:= I; MSJP1:= H "END"
    "END" "ELSE"
      "BEGIN" S[J]:=SJP1; SJP1:=S[J+1]; MSJP1:= ABSE "END";
  HE:=-HE;LOW:=UP+2;
"END" FOR J; SN:= S[N]; S[N]:= SJP1;

HI:=ABSMAXVEC(I,0,S0-1,G);
HJ:=ABSMAXVEC(J,SN+1,M,G);
"IF" J > M "THEN" J:=M;
"IF" HI > HJ "THEN"
"BEGIN" "IF" HI > MAX "THEN" "BEGIN" MAX:= HI; INDEXMAX:= I "END";
  "IF" SIGN(G[I]) = SIGN(G[S[0]]) "THEN"
    "BEGIN" "IF" HI > ABS(G[S[0]]) "THEN"
      "BEGIN" S[0]:= I;
      "IF" G[J]/G[S[N]] > 1 "THEN" S[N]:=J
    "END"
  "END" "ELSE"
    "IF" HI > ABS(G[S[N]]) "THEN"
      "BEGIN" S[N]:= "IF" G[J]/G[S[NM1]] > 1 "THEN" J "ELSE" S[NM1];
      "FOR" K:= NM1 "STEP" -1 "UNTIL" 1 "DO" S[K]:= S[K-1];
      S[0]:= I
    "END"
  "END" "ELSE"
    "BEGIN" "IF" HJ > MAX "THEN" "BEGIN" MAX:= HJ; INDEXMAX:= J "END";
    "IF" SIGN(G[J]) = SIGN(G[S[N]]) "THEN"
      "BEGIN" "IF" HJ > ABS(G[S[N]]) "THEN"
        "BEGIN" S[N]:= J; "IF" G[I]/G[S[0]] > 1 "THEN" S[0]:=I "END"
      "END" "ELSE"
        "IF" HJ > ABS(G[S[0]]) "THEN"
          "BEGIN" S[0]:= "IF" G[I]/G[S[1]] > 1 "THEN" I "ELSE" S[1];
          "FOR" K:= 1 "STEP" 1 "UNTIL" NM 1 "DO" S[K]:= S[K+1];
          S[N]:= J
        "END"
      "END"
    "END"
  "END"
"END" RANDGEBIEDEN;
EM[1]:=MAX;
"END" SNOREMEZ;
"EOP"

```



```

"CODE" 36022;
"PROCEDURE" MINMAXPOL(N,M,Y,FY,CO,EM);
"VALUE" N,M;"INTEGER" N,M;
"ARRAY" Y,FY,CO,EM;
"COMMENT" MINMAXPOL CALCULATES THE COEFFICIENTS,
CO[I],I=,...,N-1 OF THE POLYNOMIAL
P(Y)=CO[0]+CO[1]*Y+...+CO[N-1]*Y**(N-1),
THAT APPROXIMATES THE DISCRETE FUNCTION FY[I],I=0,...,M,
GIVEN FOR THE ARGUMENTS Y[I],I=0,...,M,
IN THE MINIMAX NORM.
THE ARGUMENTS MUST BE GIVEN IN MONOTONE INCREASING ORDER.
IN ARRAY EM[0:3], ONE MUST GIVE THE MAXIMUM ALLOWED NUMBER OF
ITERATIONS,EM[2].
MOREOVER,
EM[0]:=THE DIFFERENCE OF THE GIVEN FUNCTION AND THE POLYNOMIAL
IN THE FIRST APPROXIMATION POINT,
EM[1]:=THE MAXIMUM OF ! P(Y[I])-FY[I]! FOR I=0,...,M,
EM[3]:=THE NUMBER OF ITERATIONS PERFORMED.
THE PROCEDURES ELMVEC,ABSMAXVEC,POL,NEWTON,NEWGRN,
INI,SNDREMEZ
ARE USED.
REFERENCE:MEINARDUS,G.(1964,CH.7),
APPROXIMATION VON FUNKTIONEN UND IHRE NUMERISCHE BEHANDLUNG;
"BEGIN" "INTEGER" NM1,K,POMK,COUNT,CNT,J,MI;
"REAL" E,ABSE,ABSEH;
"INTEGER" "ARRAY" S[0:N];
"ARRAY" X,B[0:N]
,G[0:M];

"PROCEDURE" ELMVEC(L,U,SHIFT,A,B,X); "CODE" 34020;

"REAL" "PROCEDURE" ABSMAXVEC(K,LOW,UP,A); "CODE" 31060;

"REAL" "PROCEDURE" POL(N,X,A); "CODE" 31040;

"PROCEDURE" NEWTON(N,X,F); "CODE" 36010;

"PROCEDURE" NEWGRN(N,X,C); "CODE" 31050;

"PROCEDURE" INI(N,M,S); "CODE" 36020;

"PROCEDURE" SNDREMEZ(N,M,S,G,EM); "CODE" 36021;

"PROCEDURE" ERRPOL(N,M,E,CO,S,Y,FY,G);
"VALUE" N,M,E;"INTEGER" N,M;
"REAL" E;
"INTEGER" "ARRAY" S;"ARRAY" CO,Y,FY,G;
"COMMENT"

```


SECTION : 7.1.3.2.1

(DECEMBER 1975)

PAGE 9
;

```

"COMMENT"ERRPOL DELIVERS THE VALUE OF
CO[0]+CO[1]*Y[I]+...+CO[N-1]*Y[I]**(N-1) - FY[I]
IN G[I] FOR I=0,1,...,M AND I NOT EQUAL S[J],J=0,1,...,N.
FOR J=0,1,...,N THEN G[S[J]]:=(-1)**J*E.
THE INTEGERS S[J],FOR J=0,1,...,N ARE A SUBSET OF 0,1,...,M;
"BEGIN" "INTEGER" J,K,NM1,SJM1,SJ,S0,UP;
NM1:=N-1;S0:=SJM1:=S[0];
G[S0]:=E;
"FOR" J:=1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" SJ:=S[J];UP:=SJ-1;
"FOR" K:= SJM1+1 "STEP" 1 "UNTIL" UP "DO"
    G[K]:=FY[K]-POL(NM1,Y[K],CO);
G[SJ]:=E:=-E;
SJM1:=SJ;
"END" J;
"FOR" K:= S0-1 "STEP"-1 "UNTIL" 0 "DO"
G[K]:=FY[K]- POL(NM1,Y[K],CO);
"FOR" K:= SJ+1 "STEP" 1 "UNTIL" M "DO"
G[K]:=FY[K]- POL(NM1,Y[K],CO);
"END" ERRPOL;

```

```

INI(N,M,S);
NM1:=N-1;
MI:=EM[2];
ABSE:= 0;
"FOR" COUNT:= 1, COUNT + 1 "WHILE" COUNT <= MI & ABSE > ABSEH "DO"
"BEGIN"
POMK:=1;
"FOR" K:= 0 "STEP" 1 "UNTIL" N "DO"
"BEGIN" X[K]:= Y[S[K]]; CO[K]:= FY[S[K]]; B[K]:= POMK;
    POMK:=-POMK "END";
NEWTON(N,X,CO); NEWTON(N,X,B);
EM[0]:=
E:= CO[N]/B[N];
ELMVEC(0,NM1,0,CO,B,-E);
NEWGRN(NM1,X,CO);
ERRPOL(N,M,E,CO,S,Y,FY,G);
SNDREMEZ(N,M,S,G,EM);
    ABSEH:=ABSE;    ABSE:=ABS(E);
CNT:=COUNT;
"END" WHILE COUNT;
EM[2]:=MI;
EM[3]:=CNT;
"END" MINMAXPOL;
"EOP"

```


SECTION : 7.2.1

(SEPTEMBER 1974)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: MATHEMATICAL CENTRE.

RECEIVED: 730831.

BRIEF DESCRIPTION:

THIS SECTION CONTAINS TWO PROCEDURES:

INI SELECTS A (SUB)SET OF INTEGERS OUT OF A GIVEN SET OF INTEGERS.

SNDREMEZ EXCHANGES AT MOST $N+1$ NUMBERS WITH NUMBERS OUT OF A REFERENCE SET.

KEYWORDS:

ARGUMENTS EXTREME VALUES,
CHEBYSHEV POLYNOMIAL,
SECOND REMEZ ALGORITHM.

SUBSECTION : INI.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

"PROCEDURE" INI(N,M,S);
"VALUE" N,M, "INTEGER" N,M "INTEGER" "ARRAY" S;

THE MEANING OF THE FORMAL PARAMETERS IS:

N,M: <ARITHMETIC EXPRESSION>;
THE NUMBER OF POINTS TO BE SELECTED EQUALS $N+1$;
THE REFERENCE SET CONTAINS THE NUMBERS $0,1,\dots,M,M \geq N$;
S: <ARRAY IDENTIFIER>;
"INTEGER" "ARRAY" S(0:N);
EXIT; THE SELECTED INTEGERS ARE DELIVERED IN S.

PROCEDURES USED: NONE.

SECTION : 7.2.1

(SEPTEMBER 1974)

PAGE 2

LANGUAGE: ALGOL 60.

METHODS AND PERFORMANCE:

THE ARGUMENTS FOR WHICH THE CHEBYSHEV POLYNOMIAL OF DEGREE N ATTAINS ITS EXTREME VALUES ON THE INTERVAL $[-1,1]$ ARE TRANSFORMED TO THE INTERVAL $[0,M]$ BY A LINEAR TRANSFORMATION, FINALLY THE NUMBERS ARE PROPERLY ROUNDED.

EXAMPLE OF USE:

```
"BEGIN" "INTEGER" "ARRAY" S[0:2];
"PROCEDURE" INI(N,M,S); "CODE" 36020;
INI(2,20,S);
OUTPUT(61, "(" ("INI SELECTS OUT OF 0,1,...,20 THE NUMBERS:") ", /,
3(B=ZDB) ") ", S[0], S[1], S[2])
"END"
```

```
INI SELECTS OUT OF 0,1,...,20 THE NUMBERS:
0    10    20
```

SUBSECTION : SNDREMEZ.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:
 "PROCEDURE" SNDREMEZ(N,M,S,G,EM);
 "VALUE" N,M; "INTEGER" N,M; "INTEGER" "ARRAY" S; "ARRAY" G,EM;

THE MEANING OF THE FORMAL PARAMETERS IS:

N,M: <ARITHMETIC EXPRESSION>;

THE NUMBER OF POINTS TO BE EXCHANGED IS SMALLER THAN OR EQUAL TO N+1; THE REFERENCE SET CONTAINS THE NUMBERS 0,1,...,M, M>=N;

S: <ARRAY IDENTIFIER>;

"INTEGER" "ARRAY" S[0:N];

ENTRY: IN S ONE MUST GIVE N+1 (STRICTLY)

MONOTONE INCREASING NUMBERS OUT OF 0,...,M;

EXIT: N+1 (STRICTLY) MONOTONE INCREASING NUMBERS OUT OF THE NUMBERS 0,1,...,M;

G: <ARRAY IDENTIFIER>;

"ARRAY" G[0:M];

ENTRY: IN ARRAY G[0:M] ONE MUST GIVE FUNCTIONVALUES;

EM: <ARRAY IDENTIFIER>;

"ARRAY" EM[0:1];

ENTRY: 0<EM[0]<=G[I], I=0,...,M;

EXIT: EM[1]:=INFINITY NORM OF ARRAY G[0:M].

SECTION : 7.2.1

(SEPTEMBER 1974)

PAGE 3

PROCEDURES USED: ABSMAXVEC=CP31060.

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE:

THE SECOND REMEZ ALGORITHM IS USED (MEINARDUS, G., (1964)).

REFERENCE:

G. MEINARDUS, APPROXIMATION OF FUNCTIONS AND THEIR NUMERICAL TREATMENT (GERMAN), SPRINGER TRACTS IN NATURAL PHILOSOPHY, VOLUME 4, 1964.

EXAMPLE OF USE:

```
"BEGIN" "ARRAY" EM[0:1], G[0:7]; "INTEGER" "ARRAY" S[0:2];
"PROCEDURE" SNDREMEZ(N, M, S, G, EM); "CODE" 36021;
G[0] := 10; G[1] := 12; G[2] := -15; G[3] := -10;
G[4] := -14; G[5] := 15; G[6] := 10; G[7] := 11;
EM[0] := 10; S[0] := 0; S[1] := 3; S[2] := 6;
OUTPUT(61, "(" ("THE NUMBERS:"), /, "(" (S[J]:"), 3(B=D), /,
      "(" (G[S[J]]:"), 3(B=DD)"),
      S[0], S[1], S[2], G[S[0]], G[S[1]], G[S[2]]);
SNDREMEZ(2, 7, S, G, EM);
OUTPUT(61, "(" (//, "(" (ARE EXCHANGED WITH:), /, "(" (S[J]:"), 3(B=D), /,
      "(" (G[S[J]]:"), 3(B=DD), //,
      "(" (THE REFERENCE SET OF FUNCTIONVALUES IS:), /, 8(B=DD)"),
      S[0], S[1], S[2], G[S[0]], G[S[1]], G[S[2]],
      G[0], G[1], G[2], G[3], G[4], G[5], G[6], G[7])
"END"
```

THE NUMBERS:

S[J]: 0 3 6
G[S[J]]: 10 -10 10

ARE EXCHANGED WITH:

S[J]: 0 2 5
G[S[J]]: 10 -15 15

THE REFERENCE SET OF FUNCTIONVALUES IS:

10 12 -15 -10 -14 15 10 11

SECTION : 7,2,1

(SEPTEMBER 1974)

PAGE 4

SOURCE TEXT(S) :

```

"CODE"36020;
"PROCEDURE" INI(N,M,S);
"VALUE" N,M;"INTEGER" N,M;
"INTEGER" "ARRAY" S;
"COMMENT" INI DELIVERS (MONOTONE) THE ROUNDED VALUES
OF THE ARGUMENTS, WHERE THE CHEBYSHEV POLYNOMIAL
OF DEGREE N (TRANSFORMED TO THE INTERVAL [0,M],  $M \geq N$ )
ATTAINS ITS MAXIMUM VALUES,
IN INTEGER ARRAY S[0:N];
"BEGIN" "INTEGER" I,J,K,L;"REAL" PIN2;
    PIN2:=ARCTAN(1)*2/N;
    K:=0;L:=N-1;J:=S[0]:=0;S[N]:=M;
    "FOR" K:=K+1 "WHILE" K < L "DO"
        "BEGIN" I:=SIN(K*PIN2)**2*M;
            J:=S[K]:="IF" I<=J "THEN" J+1 "ELSE" I;
            S[L]:=M-J;L:=L-1
        "END" K;
    "IF" L*2=N "THEN" S[L]:=M/2;
    "END" INI;
    "EOP"

"CODE"36021;
"PROCEDURE" SNDREMEZ(N,M,S,G,EM);
"VALUE" N,M;"INTEGER" N,M;
"INTEGER" "ARRAY" S;"ARRAY" G,EM;
"COMMENT" SNDREMEZ EXCHANGES AT MOST N+1 NUMBERS , GIVEN IN
INTEGER ARRAY S[0:N], WITH NUMBERS OUT OF THE
REFERENCE SET 0,...,M, UNDER THE CONDITIONS:
I. THE ALTERNANCE PROPERTY OF THE FUNCTION VALUES G[S[J]],
J=0,...,N IS PRESERVED.
II.  $|G[S[J]]| \geq |EM[0]|$ , J=0,...,N.
III. THE FIRST INDEX K , WITH G[K]=INFINITY NORM OF G,
IS ONE OF THE RESULTING NUMBERS S[0],...,S[N].
IN ARRAY G[0:M] ONE MUST GIVE ERROR FUNCTION VALUES.
MOREOVER,
EM[1]:=INFINITY NORM OF G,
THE PROCEDURE ABSSMAXVEC IS USED;
"BEGIN" "INTEGER" SO,SN,SJP1,I,J,K,UP,INDEXMAX,LOW,NM1;
    "REAL" MAX,MSJP1,HI,HJ,HE,ABSE,H;
    "REAL" "PROCEDURE" ABSSMAXVEC(K,LOW,UP,A); "CODE" 31060;

```

"COMMENT"


```

INDEX MAX:=S0:=SJP1:=S[0];
HE:=EM[0];LOW:=S0+1;
MAX:=MSJP1:=ABSE:=ABS(HE);
NM1:=N-1;
"FOR" J:= 0 "STEP" 1 "UNTIL" NM1 "DO"
"BEGIN"
  UP:= S[J+1]-1;
  H:= ABSMAXVEC(I,LOW,UP,G);
  "IF" H > MAX "THEN" "BEGIN" MAX:= H; INDEX MAX:= I "END";
  "IF" H > ABSE "THEN"
    "BEGIN" "IF" HE * G[I] > 0 "THEN"
      "BEGIN" S[J]:= "IF" MSJP1 < H "THEN" I "ELSE" SJP1;
      SJP1:= S[J+1]; MSJP1:= ABSE
    "END" "ELSE"
      "BEGIN" S[J]:= SJP1; SJP1:= I; MSJP1:= H "END"
    "END" "ELSE"
      "BEGIN" S[J]:=SJP1; SJP1:=S[J+1]; MSJP1:= ABSE "END";
  HE:=-HE;LOW:=UP+2;
"END" FOR J; SN:= S[N]; S[N]:= SJP1;

HI:=ABSMAXVEC(I,0,S0-1,G);
HJ:=ABSMAXVEC(J,SN+1,M,G);
"IF" J > M "THEN" J:=M;
"IF" HI > HJ "THEN"
"BEGIN" "IF" HI > MAX "THEN" "BEGIN" MAX:= HI; INDEXMAX:= I "END";
  "IF" SIGN(G[I]) = SIGN(G[S[0]]) "THEN"
    "BEGIN" "IF" HI > ABS(G[S[0]]) "THEN"
      "BEGIN" S[0]:= I;
      "IF" G[J]/G[S[N]] > 1 "THEN" S[N]:=J
    "END"
  "END" "ELSE"
    "IF" HI > ABS(G[S[N]]) "THEN"
      "BEGIN" S[N]:= "IF" G[J]/G[S[NM1]] > 1 "THEN" J "ELSE" S[NM1];
      "FOR" K:= NM1 "STEP" -1 "UNTIL" 1 "DO" S[K]:= S[K+1];
      S[0]:= I
    "END"
  "END" "ELSE"
    "BEGIN" "IF" HJ > MAX "THEN" "BEGIN" MAX:= HJ; INDEXMAX:= J "END";
    "IF" SIGN(G[J]) = SIGN(G[S[N]]) "THEN"
      "BEGIN" "IF" HJ > ABS(G[S[N]]) "THEN"
        "BEGIN" S[N]:= J; "IF" G[I]/G[S[0]] > 1 "THEN" S[0]:=I "END"
      "END" "ELSE"
        "IF" HJ > ABS(G[S[0]]) "THEN"
          "BEGIN" S[0]:= "IF" G[I]/G[S[1]] > 1 "THEN" I "ELSE" S[1];
          "FOR" K:= 1 "STEP" 1 "UNTIL" NM 1 "DO" S[K]:= S[K+1];
          S[N]:= J
        "END"
      "END"
    "END" RANDGEBIEDEN;
    EM[1]:=MAX;
"END" SNDREMEZ;
"EOB"

```


SECTION : 7.2.3

(SEPTEMBER 1974)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: MATHEMATICAL CENTRE.

RECEIVED: 730618.

BRIEF DESCRIPTION:

MINMAXPOL CALCULATES THE COEFFICIENTS OF THE POLYNOMIAL (IN THE GRUNERT FORM) THAT APPROXIMATES A FUNCTION, GIVEN FOR DISCRETE ARGUMENTS, IN SUCH A WAY THAT THE INFINITY NORM OF THE ERROR VECTOR IS MINIMISED.

KEYWORDS:

MINIMAX POLYNOMIAL APPROXIMATION,
SECOND REMEZ EXCHANGE ALGORITHM.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:
"PROCEDURE"MINMAXPOL(N,M,Y,FY,CO,EM);
"VALUE"N,M;"INTEGER"N,M;"ARRAY"Y,FY,CO,EM;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;
THE DEGREE OF THE APPROXIMATING POLYNOMIAL IS N-1;
M: <ARITHMETIC EXPRESSION>;
THE NUMBER OF REFERENCE FUNCTION VALUES VIZ. ARGUMENTS IS M+1;
Y,FY: <ARRAY IDENTIFIERS>;
"ARRAY"Y,FY[0:M];
ENTRY: FY[I] IS THE FUNCTION VALUE AT Y[I], FOR I=0,...,M;
CO: <ARRAY IDENTIFIER>;
"ARRAY"CO[0:N];
EXIT: THE COEFFICIENTS OF THE APPROXIMATING POLYNOMIAL, OF DEGREE N-1, ARE DELIVERED IN CO[0:N-1] (CO[N-1] IS COEFFICIENT OF Y**(N-1));
EM: <ARRAY IDENTIFIER>;
"ARRAY"EM[0:3];
ENTRY: EM[2]: THE MAXIMUM ALLOWED NUMBER OF ITERATIONS;
EXIT: EM[0]: THE DIFFERENCE OF THE GIVEN FUNCTION AND THE POLYNOMIAL IN THE FIRST APPROXIMATION POINT;
EM[1]: THE INFINITY NORM OF THE ERROR OF APPROXIMATION OVER THE DISCRETE INTERVAL;
EM[3]: THE NUMBER OF ITERATIONS PERFORMED.

SECTION : 7.2.3

(SEPTEMBER 1974)

PAGE 2

PROCEDURES USED:

ELMVEC =CP34020,
 ABSMAXVEC=CP31060,
 NEWTON =CP36010,
 POL =CP31040,
 NEWGRN =CP31050,
 INI =CP36020,
 SNDREMEZ =CP36021.

REQUIRED CENTRAL MEMORY : ONE INTEGER ARRAY AND TWO REAL ARRAYS OF ORDER $N + 1$, AND ONE REAL ARRAY OF ORDER $M + 1$ ARE USED.

RUNNING TIME:

THE SECOND REMEZ ALGORITHM(ON A DISCRETE SET) IS QUADRATIC CONVERGENT; IN EACH ITERATION THE NUMBER OF OPERATIONS (MULTIPLICATIONS AND ADDITIONS) IS PROPORTIONAL TO $M \cdot N$.

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE: SEE REF. [1], CH. 7.

REFERENCES:

- [1] G. MEINARDUS.
 APPROXIMATION OF FUNCTIONS AND THEIR NUMERICAL TREATMENT
 (GERMAN). SPRINGER TRACTS IN NATURAL PHILOSOPHY,
 VOLUME 4, 1964.

EXAMPLE OF USE:

```
"BEGIN" "INTEGER" N;
"PROCEDURE" MINMAXPOL(N,M,Y,FY,CO,EM); "CODE" 36022;

"PROCEDURE" COMPUTE(N,A,B,F);
"VALUE" N,A,B;"INTEGER" N;"REAL" A,B;
"REAL" "PROCEDURE" F;
"BEGIN" "INTEGER" K,L,M;
"REAL" R,T,IDM;
"ARRAY" COEF[0:N],EM[0:3];
EM[2]:=10*N;
M:=100*N;
"BEGIN" "ARRAY" Y,FY[0:M];
IDM:=(B-A)/M;
R:=Y[0]:=A;FY[0]:=F(R);
R:=Y[M]:=B;FY[M]:=F(R);
L:=M-1;
"FOR" K:=1 "STEP" 1 "UNTIL" L "DO"
"BEGIN" R:=Y[K]:=A+K*IDM;FY[K]:=F(R) "END";
MINMAXPOL(N,M,Y,FY,COEF,EM);
OUTPUT(61, "(" ("COEF:") "/" );
"FOR" K:=0 "STEP" 1 "UNTIL" N-1 "DO" OUTPUT(61, "(" " ", COEF[K]);
OUTPUT(61, "(" /8S/, 2(N), 2(B+3ZDB), "/" ); " ("EM[0:3]" )", EM[0], EM[1],
EM[2], EM[3]);
"END";
"END" COMPUTE;

"REAL" "PROCEDURE" F(X); "VALUE" X; "REAL" X;
F:=1/(X-10);

"FOR" N:=2 "DO"
"BEGIN" OUTPUT(61, "(" //, "(" "DEGREE=" )", D//)"", N=1);
COMPUTE(N,-1,1,F)
"END"
"END";
```

DEGREE=1

```
COEF;
-1,0050378153393"-001 -1,0101010101010"-002
EM[0:3]
-5,0631947616870"-004 +5,0631947616870"-004 +20 +3
```


SOURCE TEXT(S):

```
"CODE" 36022;
"PROCEDURE" MINMAXPOL(N,M,Y,FY,CO,EM);
"VALUE" N,M;"INTEGER" N,M;
"ARRAY" Y,FY,CO,EM;
"COMMENT" MINMAXPOL CALCULATES THE COEFFICIENTS,
CO[I], I=0,...,N-1 OF THE POLYNOMIAL
P(Y)=CO[0]+CO[1]*Y+...+CO[N-1]*Y**(N-1),
THAT APPROXIMATES THE DISCRETE FUNCTION FY[I], I=0,...,M,
GIVEN FOR THE ARGUMENTS Y[I], I=0,...,M,
IN THE MINIMAX NORM.
THE ARGUMENTS MUST BE GIVEN IN MONOTONE INCREASING ORDER.
IN ARRAY EM[0:3], ONE MUST GIVE THE MAXIMUM ALLOWED NUMBER OF
ITERATIONS, EM[2].
MOREOVER,
EM[0]:=THE DIFFERENCE OF THE GIVEN FUNCTION AND THE POLYNOMIAL
IN THE FIRST APPROXIMATION POINT,
EM[1]:=THE MAXIMUM OF |P(Y[I])-FY[I]| FOR I=0,...,M,
EM[3]:=THE NUMBER OF ITERATIONS PERFORMED.
THE PROCEDURES ELMVEC, ABSMAXVEC, POL, NEWTON, NEWGRN,
INI, SNDREMEZ
ARE USED.
REFERENCE: MEINARDUS, G. (1964, CH. 7),
APPROXIMATION VON FUNKTIONEN UND IHRE NUMERISCHE BEHANDLUNG;
"BEGIN" "INTEGER" NM1,K,POMK,COUNT,CNT,J,MI;
"REAL" E,ABSE,ABSEH;
"INTEGER" "ARRAY" S[0:N];
"ARRAY" X,B[0:N],
G[0:M];

"PROCEDURE" ELMVEC(L,U,SHIFT,A,B,X); "CODE" 34020;

"REAL" "PROCEDURE" ABSMAXVEC(K,LOW,UP,A); "CODE" 31060;

"REAL" "PROCEDURE" POL(N,X,A); "CODE" 31040;

"PROCEDURE" NEWTON(N,X,F); "CODE" 36010;

"PROCEDURE" NEWGRN(N,X,C); "CODE" 31050;

"PROCEDURE" INI(N,M,S); "CODE" 36020;

"PROCEDURE" SNDREMEZ(N,M,S,G,EM); "CODE" 36021;

"PROCEDURE" ERRPOL(N,M,E,CO,S,Y,FY,G);
"VALUE" N,M,E;"INTEGER" N,M;
"REAL" E;
"INTEGER" "ARRAY" S;"ARRAY" CO,Y,FY,G; "COMMENT"
```



```
"COMMENT"ERRPOL DELIVERS THE VALUE OF
CO[0]+CO[1]*Y[I]+...+CO[N-1]*Y[I]**(N-1) = FY[I]
IN G[I] FOR I=0,1,...,M AND I NOT EQUAL S[J],J=0,1,...,N.
FOR J=0,1,...,N THEN G[S[J]]:=(+1)**J*E.
THE INTEGERS S[J],FOR J=0,1,...,N ARE A SUBSET OF 0,1,...,M;
"BEGIN" "INTEGER" J,K,NM1,SJM1,SJ,S0,UP;
NM1:=N-1;S0:=SJM1:=S[0];
G[S0]:=E;
"FOR" J:=1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" SJ:=S[J];UP:=SJ-1;
"FOR" K:= SJM1+1 "STEP" 1 "UNTIL" UP "DO"
  G[K]:=FY[K]-POL(NM1,Y[K],CO);
G[SJ]:=E:=E;
SJM1:=SJ;
"END" J;
"FOR" K:= S0-1 "STEP"-1 "UNTIL" 0 "DO"
  G[K]:=FY[K]- POL(NM1,Y[K],CO);
"FOR" K:= SJ+1 "STEP" 1 "UNTIL" M "DO"
  G[K]:=FY[K]- POL(NM1,Y[K],CO);
"END" ERRPOL;
```

```
INI(N,M,S);
NM1:=N-1;
MI:=EM[2];
ABSE:= 0;
"FOR" COUNT:= 1, COUNT + 1 "WHILE" COUNT <= MI & ABSE > ABSEH "DO"
"BEGIN"
POMK:=1;
"FOR" K:= 0 "STEP" 1 "UNTIL" N "DO"
"BEGIN" X[K]:= Y[S[K]]; CO[K]:= FY[S[K]]; B[K]:= POMK;
  POMK:=-POMK "END";
  NEWTON(N,X,CO); NEWTON(N,X,B);
  EM[0]:=
  E:= CO[N]/B[N];
  ELMVEC(0,NM1,0,CO,B,-E);
  NEWGRN(NM1,X,CO);
  ERRPOL(N,M,E,CO,S,Y,FY,G);
  SNDREMEZ(N,M,S,G,EM);
  ABSEH:=ABSE; ABSE:=ABS(E);
  CNT:=COUNT;
"END" WHILE COUNT;
EM[2]:=MI;
EM[3]:=CNT;
"END" MINMAXPOL;
  "EOP"
```